
ABSTRACT

Although the Shewhart chart serves well as the fundamental
tool for statistical process control (SPC) applications, its as-
sumptions are challenged by many modern manufacturing
environments. For example, when standard control limits
are used in applications where the process is sampled fre-
quently, autocorrelation in the measurements can result in
too many out-of-control signals. Other situations considered
in this paper include the presence of multiple components
of variation, the issue of short run process control, and the
effect of nonnormal process data. While some of these prob-
lems are subjects of ongoing research, statistical models
that extend the basic Shewhart model can provide effec-
tive solutions. This paper demonstrates the use of SAS
procedures for statistical modeling in conjunction with the
SHEWHART procedure.

INTRODUCTION

One of the main outcomes of the modern industrial qual-
ity movement has been the widespread use of statistical
process control methods to eliminate special causes of vari-
ation in manufacturing processes and to reduce common
causes of variation. Companies throughout the world have
adopted the Shewhart chart as their main SPC method, and
Shewhart charting is now taught and applied on a large
scale in both discrete components (parts) and continuous
process industries.

While the Shewhart chart is remarkably versatile, it is not
the best solution for every SPC application. There is a
growing awareness among quality engineers that standard
control charts are inappropriate or of limited value in a
number of manufacturing situations, and questions such as
the following are being asked:

� “What can I do about autocorrelation in my process
data?”

� “Is there a way to adjust control charts for multiple
sources of variation?”

� “How can I do short run process control?”
� “Are the standard control limits appropriate for non-

normal data?”

Most of these questions are subjects of current research
and debate, and it is not the goal of this paper to provide
definitive solutions. Instead the purpose is to describe some
of the more basic approaches that have been proposed

and indicate how they can be implemented with short SAS
programs.

A common theme in the sections that follow is the use
of SAS procedures for statistical modeling in conjunction
with the SHEWHART procedure for control charting. This
tandem is highly effective in a variety of nonstandard SPC
applications. The example SAS code provided in this paper
is readily extended to SAS programs that can handle large
numbers of processes. Likewise, the examples can be
incorporated in customized point-and-click interfaces devel-
oped with SAS/AF software.

AUTOCORRELATION IN PROCESS DATA

Autocorrelation has long been recognized as a natural phe-
nomenon in process industries where parameters such as
temperature and pressure vary slowly relative to the rate at
which they are measured. Only in recent years has autocor-
relation become an issue in SPC applications, particularly
in parts industries where autocorrelation is viewed as a
problem that can undermine the interpretation of Shewhart
charts. One reason for this concern is that, as automated
data collection becomes prevalent in parts industries, pro-
cesses are sampled more frequently and it is possible to
recognize autocorrelation that was previously undetected.
Another reason, noted by Box and Kramer (1992), is that the
distinction between parts and process industries is becom-
ing blurred in areas such as computer chip manufacturing.
For two other very readable discussions of this issue, refer
to Schneider and Pruett (1994) and Woodall (1993).

The standard Shewhart analysis of individual measure-
ments assumes that the process operates with a constant
mean �, and that xt (the measurement at time t) can be
represented as

xt = �+ �t

where �t is a random displacement or error from the process
mean �. The errors are typically assumed to be statistically
independent in the derivation of the conventional control
limits that are displayed at three standard deviations above
and below the center line, which represents an estimate for
�.

When Shewhart charts are constructed for measurements
that are autocorrelated, the result can be too many false sig-
nals, and users sometimes comment that “the control limits
are too tight.” This situation is illustrated in Figure 1, which
displays an individual measurement and moving range chart
for 100 observations of a chemical process.

Figure 1. Conventional Shewhart Chart

The measurements are saved in a SAS data set named
CHEMICAL. A partial listing of the measurements� is shown
in Output 1. The measurements are given by the variable
XT, and the times are given by the variable T.

Output 1. SAS Data Set CHEMICAL

Chemical Data

T XT

1 96
2 90
3 89
4 89
. .
. .
. .

96 81
97 83
98 83
99 86
100 88

The chart in Figure 1 is created with the IRCHART statement
in the SHEWHART procedure as follows:

title ’Individual Measurements Chart’;
proc shewhart graphics data=chemical;

irchart xt*t / npanel = 100
needles
split = ’/’;

label xt = ’Observed/Mvg Rng’;
run;

This is a standard application of the SHEWHART proce-
dure.y Note that the NEEDLES option is used to depict

�These are patterned after the values plotted in Figure 1
of Montgomery and Mastrangelo (1991).

yRefer to SAS Institute Inc. (1989b, 1994) for addi-
tional standard examples of the SHEWHART procedure in
SAS/QC software. Also refer to SAS Institute Inc. (1991b)
for an introduction to the SQC Menu System, which is in-
cluded with SAS/QC software and provides a point-and-click
interface to the SHEWHART procedure that is intended for
standard applications.

the process measurements as vertical displacements from
a common mean.

Diagnosing and Modeling Autocorrelation

You can diagnose autocorrelation with an autocorrelation
plot created with the ARIMA procedure:

proc arima data=chemical;
identify var = xt;

run;

Refer to SAS Institute Inc. (1993) for details on the ARIMA
procedure. The plot, shown in Output 2, indicates that the
data are highly autocorrelated with a lag one autocorrelation
of 0.83.

Output 2. Autocorrelation Plot for Chemical Data

Autocorrelations

Lag Covar Corr -1 9 8 7 6 5 4 3 2 1 0 1 2 3 4 5 6 7 8 9 1
0 48.3484 1.00000 | |********************|
1 40.1419 0.83026 | . |***************** |
2 34.7322 0.71837 | . |************** |
3 29.9509 0.61948 | . |************ |
4 24.7395 0.51169 | . |********** |
5 20.5944 0.42596 | . |********* |
6 18.4277 0.38114 | . |********. |
7 17.4002 0.35989 | . |******* . |
8 17.6213 0.36446 | . |******* . |
9 18.3638 0.37982 | . |******** . |

10 16.754 0.34653 | . |******* . |
11 16.8449 0.34841 | . |******* . |
12 17.1372 0.35445 | . |******* . |
13 16.8841 0.34922 | . |******* . |
14 17.928 0.37081 | . |******* . |
15 16.8019 0.34752 | . |******* . |

"." marks two standard errors

The partial autocorrelation plot in Output 3 suggests that
the data can be modeled with a first-order autoregressive
model, commonly referred to as an AR(1) model:

~xt � xt � � = �0 + �1~xt�1 + �t

Output 3. Partial Autocorrelation Plot

Partial Autocorrelations

Lag Correlation -1 9 8 7 6 5 4 3 2 1 0 1 2 3 4 5 6 7 8 9 1
1 0.83026 | . |***************** |
2 0.09346 | . |** . |
3 0.00385 | . | . |
4 -0.07340 | . *| . |
5 -0.00278 | . | . |
6 0.09013 | . |** . |
7 0.08781 | . |** . |
8 0.10327 | . |** . |
9 0.07240 | . |* . |

10 -0.11637 | . **| . |
11 0.08210 | . |** . |
12 0.07580 | . |** . |
13 0.04429 | . |* . |
14 0.11661 | . |** . |
15 -0.10446 | . **| . |

You can fit this model with the ARIMA procedure:

proc arima data=chemical;
identify var = xt;
estimate p = 1 method = ml;

run;

The results are shown in Output 4. The equation of the fitted
model is

~xt = 13:05+ 0:847~xt�1

Output 4. Fitted AR(1) Model

Maximum Likelihood Estimation

Approx.
Parameter Estimate Std Error T Ratio Lag
MU 85.28375 2.32973 36.61 0
AR1,1 0.84694 0.05221 16.22 1

Constant Estimate = 13.0532881

Variance Estimate = 14.2767606
Std Error Estimate = 3.77846008
AIC = 552.894156
SBC = 558.104497
Number of Residuals= 100

Strategies

There is considerable disagreement on how to handle auto-
correlation in process data. At one extreme, Wheeler (1991)
argues that the usual control limits are contaminated “only
when the autocorrelation becomes excessive (say 0.80 or
larger)” and that, in these situations, the “running record
[is] very coherent and therefore very easy to understand.”
Wheeler (1991b) concludes that “one need not be overly
concerned about the effects of autocorrelation upon the
control chart.”

A second view advocates removing autocorrelation from the
data and constructing a Shewhart chart (or an EWMA chart
or a cusum chart) for the residuals; refer, for example, to
Alwan and Roberts (1988). In the chemical data example,
the residuals can be computed as forecast errors and saved
in an output SAS data set with the FORECAST statement
in the ARIMA procedure:

proc arima data=chemical;
identify var=xt;
estimate p=1 method=ml;
forecast out=results id=t;

run;

The output data set (named RESULTS) saves the one-step-
ahead forecasts as a variable named FORECAST, and it
also contains the original variables XT and T. You can
create a Shewhart chart for the residuals by using the data
set RESULTS as input to the SHEWHART procedure:

title ’Residual Analysis Using AR(1) Model’;
proc shewhart data=results graphics;

xchart xt*t / trendvar = forecast
xsymbol = xbar
npanel = 100
ypct1 = 40
split = ’/’
nolegend;

label xt=’Residual/Forecast’;
run;

The result is displayed in Figure 2. The lower chart plots the
values of FORECAST, and the upper chart plots the resid-
uals (XT - FORECAST) together with their 3� limits. Note

that the TRENDVAR= option is needed to specify this ar-
rangement (otherwise, a standard individual measurements
chart for XT would be produced by default).

Figure 2. Residuals from AR(1) Model

The upper chart in Figure 2 resembles Figure 2 of Mont-
gomery and Mastrangelo (1991), who conclude that the
process is in control.

In addition to the residual control chart, Montgomery
and Mastrangelo (1991) suggest fitting an exponentially
weighted moving average (EWMA) model to the data and
using this model as the basis for a display that they refer to
as an EWMA center line control chart.

Recall that the EWMA statistic plotted on a conventional
EWMA control chart is defined as

zt = �xt + (1� �)zt�1

The EWMA chart (which you can construct with the MA-
CONTROL procedure) is based on the assumption that the
observations xt are independent. However, in the context of
autocorrelated process data (and more generally in time se-
ries analysis), the EWMA statistic zt plays a different role�:
it is the optimal one-step-ahead forecast for a process that
can be modeled by an ARIMA(0,1,1) model

xt = xt�1 + �t � ��t�1

provided that the weight parameter � is chosen as � = 1��.
This statistic is also a good predictor when the process can
be described by a subset of ARIMA models for which the
process is “positively autocorrelated and the process mean
does not drift too quickly.”y

You can fit an ARIMA(0,1,1) model to the chemical data
with the following statements:

�For a discussion of these roles refer to Hunter (1986).
yRefer to Montgomery and Mastrangelo (1991) and the

discussion that follows their paper.

proc arima data=chemical;
identify var=xt(1);
estimate q=1 method=ml noint;
forecast out=ewma id=t;

run;

A summary of the fitted model is shown in Output 5.

Output 5. Fitted ARIMA(0,1,1) Model

Maximum Likelihood Estimation

Approx.
Parameter Estimate Std Error T Ratio Lag
MA1,1 0.15041 0.10021 1.50 1

Variance Estimate = 14.970245
Std Error Estimate = 3.86914009
AIC = 549.868021
SBC = 552.46314
Number of Residuals= 99

The forecast values and their standard errors, together with
the original measurements, are saved in a data set named
EWMA. The following statements create a Shewhart chart
for the residuals from the fitted ARIMA(0,1,1) model by using
EWMA as an input data set for the SHEWHART procedure:

data ewma;
set ewma(firstobs=2 obs=100);

title ’Residual Analysis Using ARIMA(0,1,1) ’
’Model’;

proc shewhart data=ewma graphics;
xchart xt*t / trendvar = forecast

xsymbol = xbar
npanel = 100
ypct1 = 40
split = ’/’
nolegend;

label xt=’Residual/Forecast’;
run;

Note the similarity between this display, shown in Figure 3,
and Figure 2.�

Figure 3. Residuals from ARIMA(0,1,1) Model

�Figure 3 resembles Figure 4 of Montgomery and Mas-
trangelo (1991).

The EWMA center line control chart plots the forecasts from
the ARIMA(0,1,1) model as the center “line,” and it uses the
standard errors of prediction to determine upper and lower
control limits. You can construct this chart with the following
statements:

data ewmatab;
length _var_ $ 8 ;
set ewma
(rename=(forecast=_mean_ xt=_subx_));

_var_ = ’xt’;
_sigmas_ = 3;
_limitn_ = 1;
_lclx_ = _mean_ - 3 * std;
_uclx_ = _mean_ + 3 * std;
_subn_ = 1;

run;

title ’EWMA Center Line Control Chart’;
proc shewhart table=ewmatab graphics;

xchart xt*t / npanel = 100
llimits = 1
xsymbol = ’Center’
nolegend;

label _subx_ = ’Observed’;
run;

The variables FORECAST and STD are the forecasts and
standard errors in the data set EWMA created earlier by
the ARIMA procedure. Note that EWMA is read by the
SHEWHART procedure as a TABLE= input data set, which
has a special structure intended for applications such as
this one in which both the statistics to be plotted and their
control limits are pre-computed.y The variables in a TABLE=
data set have reserved names beginning and ending with
the underscore character, and for this reason FORECAST
and XT are temporarily renamed as MEAN and SUBX ,
respectively.

The EWMA center line chart is shown in Figure 4.z

Figure 4. EWMA Moving Center Line Chart

yRefer to the Appendix for further discussion of input
data sets for the SHEWHART procedure.

zFigure 4 is similar to Figure 5 of Montgomery and
Mastrangelo (1991).

Again, the conclusion is that the process is in control.
While Figure 2 and Figure 4 are not the only displays that
can be considered for analyzing the chemical data, their
construction illustrates the conjunctive use of the ARIMA
and SHEWHART procedures in control chart applications.

A third view of autocorrelation in process data treats depen-
dent structure as a phenomenon to be exploited rather than
removed. This approach, referred to as automatic process
control (APC) or engineering process control, has a long
history in process industries where it is common practice for
chemical engineers to devise feedback control algorithms
that continuously respond to deviations from target. In con-
trast to SPC, which assumes that the process remains on
target unless an unexpected but removable cause occurs,
APC assumes that the process is changing dynamically
due to known causes that cannot be eliminated. Instead
of avoiding “overcontrol” and “tampering,” which have a
negative connotation in the SPC framework, APC advo-
cates continuous tuning of the process to achieve minimum
variance control.

The APC approach makes considerable use of time se-
ries models, not only for describing process disturbances
but also for describing the relationships between tuning
(compensation) variables and output variables, and it also
involves optimal control theory. The basic idea is to fore-
cast the location of the process at the next time step and
then determine an adjustment that brings the process to
target. Excellent descriptions of this approach and consid-
erable discussion of the differences between APC and SPC
are provided by a number of authors, including MacGre-
gor (1987,1990), MacGregor, Hunter, and Harris (1988),
Montgomery et al. (1994), and Box and Kramer (1992).

Summary

There are no simple answers to the question of how to
deal with autocorrelation in process data, and this issue is
a research topic with divergent views. However, in many
situations the ARIMA procedure is a key tool for identifying
autocorrelation and modeling process disturbances, and it
can be used in conjunction with the SHEWHART procedure
to create a variety of useful control charts.

MULTIPLE COMPONENTS OF VARIATION

In the preceding section, the excessive variation observed
in the conventional Shewhart chart displayed in Figure 1
is the result of positive autocorrelation in the data. The
variation is “excessive” not because it is due to special
causes of variation but because the Shewhart model, which
provides the basis for the control limits, is inappropriate.
This section considers another form of departure from the
Shewhart model; here, measurements are independent
from one subgroup sample to the next, but there are multiple
components of variation for each measurement. This is
illustrated with an example involving two components.�

�Also refer to Chapter 5 of Wheeler and Chambers
(1986) for a lucid explanation of the effects of subgrouping
and sources of variation on control charts.

A company that manufactures polyethylene film is con-
cerned with monitoring statistical control of an extrusion
process that produces a continuous sheet of film. At peri-
odic intervals of time, samples are taken at four locations
(referred to as lanes) along a cross section of the sheet,
and a test measurement is made of each sample. The test
values are saved in a SAS data set named FILM. A partial
listing of FILM is shown in Output 6.

Output 6. SAS Data Set FILM

Polyethylene Sheet Measurements

SAMPLE LANE TESTVAL

1 A 93
1 B 87
1 C 92
1 D 78
2 A 87
2 B 83
2 C 79
2 D 77
. . .
. . .
. . .

As a preliminary step in the analysis, the data are sorted by
lane and visually screened for outliers (test values greater
than 130) with box plots created as follows:

proc sort data=film;
by lane;

run;

title ’Outlier Analysis’;
proc shewhart data=film graphics;

boxchart testval*lane /
boxstyle=schematicid
stddevs
nolimits
hoffset=5
idsymbol=dot
cboxfill=red
nolegend
vref=130
vreflab=’Outlier Cutoff’;

id sample;
run;

The BOXSTYLE= option requests schematic box plots with
outliers identified by the value of the ID variable SAMPLE.
The NOLIMITS option suppresses the control limits for lane
means that are displayed by default. The STDDEVS option
specifies that the estimate of the process standard deviation
is to be based on subgroup standard deviations rather than
ranges. Although this estimate is not needed here because
control limits are not displayed, it is recommended that you
specify STDDEVS whenever you are working with subgroup
sample sizes greater than ten. The VREF= and VREFLAB=
options specify the reference line and label positioned at
the value of 130, and the HOFFSET= option requests the
offsets at both ends of the horizontal axis.

The display is shown in Figure 5.

Figure 5. Outlier Analysis for FILM Data

Figure 6 shows similarly created box plots for the data in
FILM after the outliers have been removed.�

Figure 6. FILM Data Without Outliers

The variation within lane can also be examined with a
comparative histogram created as follows:

title ’Data Without Outliers’;
proc capability data=film graphics noprint;

comphistogram testval /
class=lane
nrows=4
intertile=1
vaxislabel =’Pct’
;

inset max=’Max’ / pos=ne ;
run;

The comparative histogram is shown in Figure 7.

�For the remainder of this section, unless otherwise
indicated, it is assumed that the outliers are deleted from
FILM.

Figure 7. Comparative Histogram

Without additional information about the process, it is tempt-
ing to create a conventionalX andR chart for the test values
grouped by the variable SAMPLE. This is a straightforward
application of the XRCHART statement in the SHEWHART
procedure:

proc sort data=film;
by sample;

title ’Shewhart Chart for Means and Ranges’;
proc shewhart data=film graphics;

xrchart testval*sample /
split = ’/’
npanel = 60
limitn = 4
nolegend
alln;

label testval=’Avg Test Value/Range’;
run;

The X and R chart is displayed in Figure 8.

Figure 8. Conventional X and R Chart

Note that the LIMITN= option in the preceding XRCHART
statement requests control limits for a nominal sample size
of four, and the ALLN option specifies that these limits
are to be displayed with all subgroup samples regardless
of sample size (not all samples have four measurements
because outliers have been removed). If LIMITN= and
ALLN were not specified, the control limits would vary with
sample size.

The number of out-of-control points in Figure 8 would ordi-
narily indicate that the process is not in statistical control.
In this situation, however, the process is known to be quite
stable, and the data have been screened for outliers. Thus,
it is suspected that the control limits are not appropriate for
the data.

Recall that the standard Shewhart analysis assumes that
sampling variation, also referred to as within-group variation,
is the only source of variation. Writing xij for the jth
measurement within the ith subgroup, you can express the
model for the conventional X and R chart as

xij = �+ �W �ij

for i = 1;2; : : : k and j = 1;2; : : : n. In this example,
there are n = 4 measurements in a subgroup and there
are k = 56 subgroup samples. The random variables �ij
are assumed to be independent with zero mean and unit
variance, and �2

W is the within-subgroup variance. The
parameter � denotes the process mean.

In a process such as film manufacturing, however, this
model is not adequate because there is additional variation
due to changes in temperature, pressure, raw material, and
other factors. Instead, a useful model is

xij = �+ �B!i + �W �ij

where �2
B is the between-subgroup variance, the random

variables !i are independent with zero mean and unit vari-
ance, and the random variables !i are independent of the
random variables �ij.�

To plot the subgroup averages

xi: �

nX
j=1

xij=n

on a control chart, you need expressions for the expectation
and variance of xi:. These are

E(xi:) = �

Var(xi:) = �
2
B + (�2

W =n)

Thus the center line should be located at �̂, and 3� limits
should be located at

�̂� 3

qc�2
B + c�2

W=n

wherec�2
B and c�2

W denote estimates of the variance compo-
nents.

�This is the notation used in Chapter 3 of Wetherill and
Brown (1991), which provides helpful discussion of this
issue.

You can use a variety of SAS procedures for fitting linear
models to estimate the variance components. The following
statements show how this can be done with the MIXED
procedure:

proc mixed data=film;
class sample;
model testval = / s;
random sample;
make ’solutionf’ out=sf;
make ’covparms’ out=cp;

run;

The results are shown in Output 7. Note that the parameter

estimates are c�2
B = 19:25, c�2

W = 39:68, and �̂ = 88:90.

Output 7. Partial Output from MIXED Procedure

Covariance Parameter Estimates (REML)

Cov Parm Ratio Estimate Std Error Z

SAMPLE 0.48516517 19.25257992 5.67829660 3.39
Residual 1.00000000 39.68252702 4.35915223 9.10

Covariance Parameter Estimates (REML)

Pr > |Z|

0.0007
0.0001

Solution for Fixed Effects

Parameter Estimate Std Error DDF T Pr > |T|

INTERCEPT 88.89629708 0.72504535 55 122.61 0.0001

The following statements merge the output data sets from
the MIXED procedure into a SAS data set named NEWLIM
that contains the appropriately derived control limit param-
etersy for the average test value:

data sf; set sf; rename _est_=est;
data cp; set cp sf; keep est;
proc transpose data=cp out=newlim;

data newlim;
set newlim;
drop _name_ _label_ col1-col3;
length _var_ _subgrp_ _type_ $8;
_var_ = ’testval’;
_subgrp_ = ’sample’;
_type_ = ’estimate’;
_limitn_ = 4;
_mean_ = col3;
_stddev_ = sqrt(4*col1 + col2);
output;

run;

Here the variable LIMITN is assigned the value of n, the
variable MEAN is assigned the value of �̂, and the variable
STDDEV is assigned the value of

d�adj �

q
4c�2

B + c�2
W

ySee the Appendix for additional discussion.

In the next set of statements the SHEWHART procedure
reads these parameter estimates and displays the X and R

chart shown in Figure 9. The control limits for the X chart
are displayed at

�̂� 3d�adj=
p
n

title ’Control Chart With Adjusted Limits’;

proc shewhart data=film limits=newlim
graphics;
xrchart testval*sample /

readlimits
npanel = 60 ;

run;

Note that the chart in Figure 9 correctly indicates that the
variation in the process is due to common causes.

Figure 9. Derived Control Limits

You can use a similar set of statements to display the
derived control limits in NEWLIM on an X and R chart for
the original data (including outliers), as shown in Figure 10.

Figure 10. Derived Control Limits With Raw Data

A simple alternative to the chart in Figure 9 is a con-
ventional “individual measurements” control chart for the
subgroup means which you can construct with the following
statements:

proc means noprint data=film;
by sample;
output out=filmx mean=lanemean;

run;

title ’Individuals Chart for Means’;
proc shewhart data=filmx graphics;

irchart lanemean*sample /
split = ’/’
npanel = 60
nochart2;

label lanemean=’Avg Value/Mvg Rng’;
run;

The result (see Figure 11) is only roughly equivalent to theX

chart in Figure 9 because the standard error for the means
in Figure 11 is determined by averaging the moving ranges
of the means and dividing by the unbiasing constant d2. If
you replace this standard error with a value determined from
the sample standard deviation of the means then the control
limits will be identical (the SAS statements are not shown
here). The variance components approach is preferable
because it yields separate estimates of the components
due to lane and sample, as well as a number of hypothesis
tests (these require assumptions of normality). However, in
applying this method you should be careful to use data that
represent the process in a state of statistical control.

Figure 11. Individuals Chart for Averages

A further advantage of using the variance components
approach is that you can easily extend your model to include
a variety of effects. In this example, if you plot the test
values by time using symbol markers to denote the lane
(see Figure 12), it appears that there is an effect due to
lane.

Figure 12. Variation within Sample

The following statements fit a mixed model in which LANE
is a fixed effect and SAMPLE is a random effect.

proc mixed data=film;
class sample lane;
model testval = lane;
random sample;
lsmeans lane;

run;

The results of this analysis are given in Output 8.

Output 8. Partial Output from MIXED Procedure

Covariance Parameter Estimates (REML)

Cov Parm Ratio Estimate Std Error Z

SAMPLE 0.82943785 22.52245916 5.66313893 3.98
Residual 1.00000000 27.15388406 3.01371956 9.01

Covariance Parameter Estimates (REML)

Pr > |Z|

0.0001
0.0001

Tests of Fixed Effects

Source NDF DDF Type III F Pr > F

LANE 3 162 26.37 0.0001

Least Squares Means

Level LSMEAN Std Error DDF T Pr > |T|

LANE A 88.04445518 0.94862578 162 92.81 0.0001
LANE B 93.22627337 0.94862578 162 98.28 0.0001
LANE C 89.89900064 0.94862578 162 94.77 0.0001
LANE D 84.60714286 0.94184795 162 89.83 0.0001

Note that the lane effect is significant, and this suggests
that it might be wise to maintain separate control charts for
each lane.

Summary

In situations involving multiple sources of variation, appro-
priate control limits for averages can be derived by fitting a

linear model to measurements taken while the process is in
statistical control and estimating the variance components.
The MIXED and GLM procedures (among others) are use-
ful for fitting the model and creating output SAS data sets
with estimates that can subsequently serve as input to the
SHEWHART procedure. This approach offers considerable
generality because it is possible to fit mixed models that
incorporate time dependent structure as well as mixed and
fixed effects.

SHORT RUN PROCESS CONTROL

The issue of short run process control, like the problem
of the previous section, requires special care in analyzing
process variability.

When conventional Shewhart charts are used to establish
statistical control, the initial control limits are typically based
on between 25 and 30 subgroup samples. However, this
amount of data is often not available in manufacturing
situations where product changeover occurs frequently or
production runs are limited.

A variety of methods have been introduced for analyzing
data from a process that is alternating between short runs
of multiple products. The methods commonly used in the
United States are variations of two basic approaches:�

� the difference from nominal approach. A product-
specific nominal value is subtracted from each mea-
sured value, and the differences (together with appro-
priate control limits) are charted. Here it is assumed
that the nominal value represents the central location
of the process (ideally estimated with historical data)
and that the process variability is constant across
products.

� the standardization approach. Each measured value
is standardized with a product-specific nominal and
standard deviation values. This approach is followed
when the process variability is not constant across
products.

These approaches are highlighted in this section because
of their popularity, but it is worth noting two alternatives that
are technically more sophisticated:

� Hillier (1969) provided a method for modifying the
usual control limits for X and R charts in startup
situations where fewer than 25 subgroup samples
are available for estimating the process mean �

and standard deviation �; also refer to Quesenberry
(1993).

� Quesenberry (1991a, 1991b) introduced the so-
called Q chart for short (or long) production runs,
which standardizes and normalizes the data using
probability integral transformations.

SAS examples illustrating these alternatives are given by
Rodriguez and Bynum (1992).

�For a review of related methods, refer to Al-Salti and
Statham (1994).

Difference From Nominal

The following example� is adapted from an application
in aircraft component manufacturing. A metal extrusion
process is used to make three slightly different models of
the same component. The three product types (labeled M1,
M2, and M3) are produced in small quantities because the
process is expensive and time-consuming.

Output 9 indicates the structure of a SAS data set named
OLD in which the diameter measurements for various short
runs have been saved. Samples 1 to 30 are to be used to
estimate the process standard deviation� for the differences
from nominal.

Output 9. Data Set OLD

Preliminary Data For Short Production Runs

PRODTYPE DIAMETER SAMPLE

M3 13.99 1
M3 14.69 2
M3 13.86 3
M3 14.32 4
M3 13.23 5
M1 17.55 6
M1 14.26 7
M1 14.62 8
M1 12.97 9
M2 16.18 10
. . .
. . .
. . .

M1 16.26 25
M3 16.68 26
M3 15.60 27
M3 14.86 28
M3 16.67 29
M3 14.35 30

In short run applications involving many product types, it is
common practice to maintain a database for the nominal
values for the product types. Here the nominal values are
saved in a SAS data set named NOMVAL, which is listed in
Output 10.

Output 10. Nominal Values

Nominal Values for Product Types

PRODTYPE NOMINAL

M1 15.0
M2 15.5
M3 14.8
M4 15.2

To compute the differences from nominal, you must merge
the data with the nominal values. You can do this with the
following SAS statements:

proc sort data=old; by prodtype;
proc sort data=nomval; by prodtype;

data old;
merge nomval old(in = a);

by prodtype; if a;
diff = diameter - nominal;

proc sort data=old; by sample;
run;

�Refer to Chapter 1 of Wheeler (1991a) for a similar
example.

Note that an IN= variable is used in the MERGE statement
to allow for the fact that NOMVAL includes nominal values
for product types that are not represented in OLD. Output 11
lists the merged data set OLD.

Output 11. Data Merged with Nominal Values

SAMPLE PRODTYPE NOMINAL DIAMETER DIFF

1 M3 14.8 13.99 -0.81
2 M3 14.8 14.69 -0.11
3 M3 14.8 13.86 -0.94
4 M3 14.8 14.32 -0.48
5 M3 14.8 13.23 -1.57
6 M1 15.0 17.55 2.55
7 M1 15.0 14.26 -0.74
8 M1 15.0 14.62 -0.38
9 M1 15.0 12.97 -2.03

10 M2 15.5 16.18 0.68
11 M2 15.5 15.29 -0.21
12 M2 15.5 16.20 0.70
13 M3 14.8 13.89 -0.91
14 M3 14.8 12.71 -2.09
15 M3 14.8 14.32 -0.48
16 M3 14.8 15.35 0.55
17 M2 15.5 15.08 -0.42
18 M2 15.5 14.72 -0.78
19 M2 15.5 14.79 -0.71
20 M2 15.5 15.27 -0.23
21 M2 15.5 15.95 0.45
22 M1 15.0 14.78 -0.22
23 M1 15.0 15.19 0.19
24 M1 15.0 15.41 0.41
25 M1 15.0 16.26 1.26
26 M3 14.8 16.68 1.88
27 M3 14.8 15.60 0.80
28 M3 14.8 14.86 0.06
29 M3 14.8 16.67 1.87
30 M3 14.8 14.35 -0.45

For the time being, assume that the variability in the process
is constant across product types. To estimate the common
process standard deviation �, you first estimate � for each
product type based on the average of the moving ranges
of the differences from nominal. You can do this in several
steps, the first of which is to sort the data and compute the
average moving range with the SHEWHART procedure as
follows:

proc sort data=old; by prodtype;

proc shewhart data=old;
irchart diff*sample /

nochart
outlimits=baselim;

by prodtype;
run;

The purpose of this procedure step is simply to save the
average moving range for each product type in the OUT-
LIMITS= data set (note that PRODTYPE is specified as
a BY variable.) In standard applications of the IRCHART
statement, the average moving range is displayed as the
center line in moving range charts that are created by de-
fault. Here the NOCHART option is used to suppress the
display of the charts (these are not valid since there are
insufficient measurements for each product type.)

Output 12 lists the data set BASELIM, which contains the
average moving ranges as values of the variable R , in
addition to other control limit information.

Output 12. Values of R by Product Type

Control Limits By Product Type

PRODTYPE _VAR_ _SUBGRP_ _TYPE_ _LIMITN_ _ALPHA_ _SIGMAS_

M1 DIFF SAMPLE ESTIMATE 2 .0026998 3
M2 DIFF SAMPLE ESTIMATE 2 .0026998 3
M3 DIFF SAMPLE ESTIMATE 2 .0026998 3

_LCLI_ _MEAN_ _UCLI_ _LCLR_ _R_ _UCLR_ _STDDEV_

-3.13153 0.12924 3.39001 0 1.22646 4.00628 1.08692
-1.78485 -0.06551 1.65382 0 0.64669 2.11243 0.57311
-3.22326 -0.19034 2.84258 0 1.14076 3.72633 1.01097

To obtain a combined estimate of �, you can use the MEANS
procedure to average the average ranges in BASELIM and
then divide by the unbiasing constant d2.

proc means data=baselim noprint;
var _r_;
output out=difflim (keep=_r_) mean=_r_;

data difflim;
set difflim;
drop _r_;
length _var_ _subgrp_ $ 8;
_var_ = ’diff’;
_subgrp_ = ’sample’;
_mean_ = 0.0;
_stddev_ = _r_ / d2(2);
_limitn_ = 2;
_sigmas_ = 3;

run;

The data set DIFFLIM is structured for subsequent use as
an input LIMITS= data set by the SHEWHART procedure
(below). The variables in a LIMITS= data set provide pre-
computed control limits or—as in this case—the parameters
from which control limits are to be computed. These vari-
ables have reserved names that begin and end with the
underscore character. Here the variable STDDEV saves
the estimate of �, and the variable MEAN saves the mean
of the differences from nominal (recall that this mean is zero
since the nominal values are assumed to represent the pro-
cess mean for each product type.) The identifier variables
VAR and SUBGRP record the names of the process and

subgroup variables (these variables are critical in applica-
tions involving many product types.) The variable LIMITN
is assigned a value of two to specify moving ranges of two
consecutive measurements, and the variable SIGMAS is
assigned a value of three to specify 3� limits. The data set
DIFFLIM is listed in Output 13.

Output 13. Estimates of Mean and Standard Deviation

Control Limit Parameters For Differences

_VAR_ _SUBGRP_ _MEAN_ _STDDEV_ _LIMITN_ _SIGMAS_

diff sample 0 0.89034 2 3

Now that the control limit parameters are saved in DIFFLIM,
you can construct short run control charts for new data as
follows. Suppose that diameters for an additional 30 parts

(numbered 31 to 60) are measured and saved in a SAS
data set named NEW, as listed in Output 14.

Output 14. Data Set NEW

New Data For Short Production Runs

SAMPLE PRODTYPE DIAMETER

31 M2 14.69
32 M2 15.39
33 M2 14.56
34 M2 15.02
35 M2 13.93
36 M1 17.55
37 M1 14.26
38 M3 14.42
39 M3 12.77
40 M3 15.48
41 M3 14.59
42 M2 16.20
43 M2 14.59
44 M2 13.41
45 M2 15.02
46 M2 16.05
47 M2 15.08
48 M2 14.72
49 M2 14.79
50 M1 14.77
51 M1 15.45
52 M1 14.28
53 M1 14.69
54 M2 15.41
55 M2 16.26
56 M2 17.38
57 M3 15.60
58 M3 14.86
59 M3 16.67
60 M3 14.35

As before, you merge the measurements in NEW with their
corresponding nominal values in NOMVAL and compute
the differences from nominal.

proc sort data=new;
by prodtype;

data new;
merge nomval new(in = a);

by prodtype; if a;
diff = diameter - nominal;

run;

To construct a short run individual measurements and mov-
ing range chart, first sort the observations in NEW by
SAMPLE, and then use the SHEWHART procedure with
DIFFLIM specified as an input LIMITS= data set.

proc sort data=new;
by sample;

title ’Chart for Difference from Nominal’;
symbol1 v=dot ;
symbol2 v=plus ;
symbol3 v=circle;

proc shewhart data=new
limits=difflim graphics;

irchart diff * sample = prodtype /
readlimits
split=’/’;

label diff = ’Difference/Mvg Rng’;
run;

The chart is displayed in Figure 13. Note that the product
types are identified with symbol markers as requested by

specifying PRODTYPE as a symbol variable after the =
sign in the IRCHART statement.

Figure 13. Short Run Control Chart

You can also identify the product types with a legend by
specifying PRODTYPE as a PHASE variable.

title ’Chart for Difference from Nominal’;

proc shewhart
data=new (rename=(prodtype=_phase_))
limits=difflim graphics;
irchart diff * sample /

readlimits
readphases=all
phaseref
phasebreak
phaselegend
split=’/’;

label diff = ’Difference/Mvg Rng’;
run;

The display is shown in Figure 14.

Figure 14. Identification of Product Types

In some applications, it may be useful to replace the moving
range chart with a plot of the nominal values. You can do this
with the TRENDVAR= option in the XCHART statement�

provided that you reset the value of LIMITN to one (this
specifies a subgroup sample of size one).

data difflim;
set difflim;
_var_ = ’diameter’;
_limitn_ = 1;

run;

title ’Differences and Nominal Values’;

proc shewhart data=new
limits=difflim graphics;

xchart diameter*sample (prodtype) /
readlimits
nolimitslegend
nolegend
split = ’/’
blockpos = 3
blocklabtype = scaled
blocklabelpos = left
xsymbol = xbar
trendvar = nominal;

label diameter = ’Difference/Nominal’
prodtype = ’Product’;

run;

The display is shown in Figure 15. Note that you identify the
product types by specifying PRODTYPE as a block variable
enclosed in parentheses after the subgroup variable SAM-
PLE. The BLOCKLABTYPE= option specifies that values of
the block variable are to be scaled (if necessary) to fit the
space available in the block legend. The BLOCKLABEL-
POS= optiony specifies that the label of the block variable
is to be displayed to the left of the block legend.

Figure 15. Short Run Control Chart with Nominal Values

�The TRENDVAR= option is not available in the IR-
CHART statement.

yAvailable in Release 6.10.

Are the Variances Constant?

The difference-from-nominal chart should be accompanied
by a test that checks whether the variances for each prod-
uct type are identical (homogeneous). Wheeler (1991a)
describes a simple range-based method. A second method,
used by a number of manufacturing companies, is to apply
a Kruskal-Wallis test to the moving ranges for each product
type. Both methods can be implemented with short SAS
programs.

This section illustrates a third method, Levene’s test of
homogeneity, which is particularly appropriate for short
run applications because it is robust to departures from
normality; refer to Snedecor and Cochran (1980). One
reason Levene’s test has not been widely used for this
purpose is that it is more complicated than the first two
methods. However, you can implement Levene’s method
by using the GLM procedure to construct a one-way analysis
of variance for the absolute deviations of the diameters from
averages within product types.

proc sort data=old; by prodtype;

proc means data=old noprint;
var diameter;
output out=oldmean (keep=prodtype diammean)

mean=diammean;
by prodtype;

data old;
merge old oldmean; by prodtype;
absdev = abs( diameter - diammean );

proc means data=old noprint;
var absdev;
by prodtype;
output out=stats

n=n mean=mean css=css std=std;

title ’Test for Constant Variance’;
proc glm data=old outstat=glmout ;

class prodtype;
model absdev = prodtype;

run;

A partial listing of the results is displayed in Output 15. The
large P -value (0.3386) indicates that the data support the
hypothesis of homogeneity.

Output 15. Levene’s Test

General Linear Models Procedure

Dependent Variable: ABSDEV

Source DF Sum of Squares F Value Pr > F

Model 2 1.02518468 1.13 0.3386

Error 27 12.27280629

Corrected Total 29 13.29799097

Standardization

When the variances across product types are not constant,

then various authors recommend standardizing the differ-
ences from nominal and displaying them on a common chart
with control limits at �3.

To illustrate this method, assume that the hypothesis of
homogeneity is rejected for the differences in OLD. Then
you can use the product-specific estimates of � in BASELIM
to standardize the differences from nominal in NEW as
follows:

proc sort data=baselim; by prodtype;
proc sort data=new; by prodtype;

data new;
keep sample prodtype z diameter

diff nominal _stddev_;
label sample = ’Sample Number’
format diff 5.2 ;
merge baselim new(in = a);

by prodtype; if a;
z = diff / _stddev_ ;

proc sort data=new; by sample;
run;

The following statements create the standardized chart:

title ’Standardized Chart’;
symbol v=dot;
proc shewhart data=new graphics;

irchart z*sample (prodtype) /
blocklabtype = scaled
mu0 = 0
sigma0 = 1
split = ’/’;

label prodtype = ’Product Classification’
z = ’Std Diff/Mvg Rng’;

run;

Note that the options MU0= and SIGMA= specify that the
control limits for the standardized differences from nominal
are to be based on the parameters � = 0 and � = 1. The
chart is displayed in Figure 16.

Figure 16. Standardized Difference Chart

Summary

The two most basic approaches to short run process control
are to chart differences from product-specific nominal values
when the variability is constant across products and to
chart standardized differences when the variability is not
constant. In practice, this requires maintaining a database
of nominal values and standard deviations for products in
addition to the process data. Charting the differences is
a straightforward application of the SAS DATA step and
the SHEWHART procedure, and the examples presented
here can be extended to any number of product types. It
is important to check whether the variances are constant
(homogeneous), and this can be done with Levene’s test by
using the GLM procedure.

NONNORMAL PROCESS DATA

A number of authors have pointed out that Shewhart charts
for subgroup means work well whether the measurements
are normally distributed or not. Refer to, for instance,
Schilling and Nelson (1976) and Wheeler (1991). On the
other hand, the interpretation of standard control charts for
individual measurements (X charts) is affected by depar-
tures from normality, and nonnormal data are not uncom-
mon in SPC applications. Two examples are coordinate
measurements that have a geometrically fixed lower (but
not upper) bound, and waiting times.

In situations involving a large number of measurements it
may be possible to subgroup the data and construct an X

chart instead of an X chart. However, the measurements
should not be subgrouped arbitrarily for this purpose.� If
subgrouping is not possible and considerable nonnormality
is evident, two alternatives are to transform the data to
normality (preferably with a simple transformation such as
the log transformation) or modify the usual limits based on
a suitable model for the data distribution.

The second of these alternatives is illustrated here with data
from a study conducted by a service center. The time taken
by staff members to answer the phone was measured, and
the delays (in minutes) were saved as values of a variable
named TIME in a SAS data set named CALLS. A partial
listing of CALLS is shown in Output 16.

Output 16. Data Set CALLS

RECNUM TIME

1 3.2
2 3.1
3 3.1
4 2.9
5 2.8
. .
. .
. .

49 2.6
50 2.9

As a first step, the delays were analyzed using an X chart
created with the following statements:

�Refer to Wheeler and Chambers (1986) for a discussion
of subgrouping.

title ’Standard Analysis of Individual ’
’Delays’;

proc shewhart data=calls graphics ;
irchart time * recnum /

rtmplot = schematic
outlimits = delaylim
cboxfill = grey
nochart2;

run;

The chart is displayed in Figure 17.

Figure 17. Standard Control Limits for Delays

It is tempting to conclude that the 41st point signals a special
cause of variation and that otherwise the process variation
is well within the control limits. However, the box plot in the
right margin (requested with the RTMPLOT= option) helps
you seey that the distribution of delays is skewed. In other
words, the reason that almost all the measurements are
grouped well within the control limits is that the limits are
incorrect—and not that the process is too good for the limits.

The control limits are saved in a SAS data set named
DELAYLIM, which is listed in Output 17.

Output 17. Control Limits Assuming Normality

Control Limits for Standard Chart

_VAR_ _SUBGRP_ _TYPE_ _LIMITN_ _ALPHA_

TIME RECNUM ESTIMATE 2 .0026998

_SIGMAS_ _LCLI_ _MEAN_ _UCLI_ _STDDEV_

3 1.77001 2.91035 4.05070 0.38011

The control limits can be modified by replacing them with

yThis assumes the process is in statistical control; oth-
erwise the box plot could not be interpreted as a repre-
sentation of the process distribution. You can check the
assumption of normality with goodness-of-fit tests by using
the CAPABILITY procedure as shown in the statements that
follow.

corresponding percentiles from a fitted lognormal distribu-
tion. The equation for the lognormal density function is

f(x) =
1

x
p

2��
exp(�[log(x)� �]

2
=2�2

) ; x > 0 ;

where � denotes the shape parameter and � denotes the
scale parameter.

The following statements use the CAPABILITY procedure�

to fit a lognormal model and superimpose the fitted density
on a histogram of the data.

title ’Lognormal Fit for Delay Distribution’;

proc capability data=calls graphics ;
histogram time /

lognormal( threshold = 2.3 )
cfill = grey
outfit = lnfit
nolegend ;

inset n = ’Number of Calls’
lognormal( sigma=’Shape’ (4.2)

zeta =’Scale’ (5.2)
theta ) /

pos = ne;
run;

The LOGNORMAL option in the HISTOGRAM statement is
used to request the lognormal fit, and the INSET statement
is used to request the boxed information displayed on the
histogram. The histogram is shown in Figure 18.

Figure 18. Distribution of Delays

Parameters of the fitted distribution and results of goodness-
of-fit tests are summarized in Output 18. The large P -values
for the goodness-of-fit tests are evidence that the lognormal
model provides a good fit.

�Refer to Rodriguez and Bynum (1993) or SAS Institute
Inc. (1994) for details concerning recent enhancements to
the CAPABILITY procedure, including the INSET statement
and additional goodness-of-fit tests based on the empirical
distribution function (EDF).

Output 18. Summary of Fitted Lognormal Model

Fitted Distribution: Lognormal

Parameters EDF Goodness-of-Fit

Threshold (Theta) 2.3 A-D (A-Square) 0.34761167
Scale (Zeta) -0.6891925 Pr > A-Square 0.4765
Shape (Sigma) 0.64121248
Est Mean 2.91655007 C-von M (W-Square) 0.05863868
Est Std Dev 0.4396814 Pr > W-Square 0.4106

Chi-Square Goodness-of-Fit Kolmogorov (D) 0.09208575
Pr > D >.15

Chi-Square 6.81723374
Df 4
Pr > Chi-Square 0.1459

This information is also saved in a SAS data set named
LNFIT, which is listed below.

Output 19. Output Data Set LNFIT

Parameters of Fitted Lognormal Model

_VAR_ _CURVE_ _LOCATN_ _SCALE_ _SHAPE1_ _MIDPTN_

TIME LNORMAL 2.3 -0.68919 0.64121 4.2

_ADASQ_ _ADP_ _CVMWSQ_ _CVMP_ _KSD_ _KSP_

0.34761 0.47648 0.058639 0.41060 0.092086 0.15

The following statements use the information in LN-
FIT to replace the control limits in DELAYLIM with lim-
its computed from percentiles of the fitted lognormal
model. The �th percentile of the lognormal distribution
is P� = exp(���1(�) + �), where ��1 denotes the inverse
standard normal cumulative distribution function.

data delaylim;
merge delaylim lnfit;
drop _sigmas_ ;
_lcli_ = _locatn_ +

exp(_scale_+probit(0.5*_alpha_)*_shape1_);
_ucli_ = _locatn_ +

exp(_scale_+probit(1-0.5*_alpha_)*_shape1_);
_mean_ = _locatn_ +

exp(_scale_+0.5*_shape1_*_shape1_);
run;

The next statements use the SHEWHART procedure to
construct anX chart using the modified limits in DELAYLIM.

title ’Lognormal Control Limits for Delays’;
proc shewhart data=calls limits=delaylim

graphics;
irchart time * recnum /

rtmplot = schematic
cboxfill = grey
nochart2
readlimits;

run;

The chart is displayed in Figure 19.

Figure 19. Adjusted Control Limits for Delays

Clearly the process is in control, and the control limits
(particularly the lower limit) are appropriate for the data.
The particular probability level � = 0:0027 associated with
these limits is somewhat immaterial, and other values of �
such as 0.001 or 0.01 could be specified with the ALPHA=
option in the original IRCHART statement.

Summary

The assumption of normality is critical to the interpretation
of control charts for individual measurements. This can
be checked with graphical displays or with goodness-of-fit
tests available with the CAPABILITY procedure. You can
also use this procedure to fit nonnormal models such as
the lognormal distribution and then derive modified control
limits from the fitted model.

APPENDIX: INPUT DATA SETS FOR THE
SHEWHART PROCEDURE

To make the most effective use of the SHEWHART pro-
cedure, you should understand the roles and structures of
the various types of SAS data sets that can be read by
the procedure. This appendix provides a brief overview
that supplements the discussion provided with the SAS ex-
amples in this paper. For complete details, refer to SAS
Institute Inc. (1989a, 1994).

DATA= Data Sets for Reading Raw Data

The simplest input data set is the DATA= data set, which you
use to provide the SHEWHART procedure with raw mea-
surements.� This data set must contain at least one process
variable, whose values are the process measurements, and
one subgroup variable, whose values determine the rational
subgroups into which the measurements are classified (or,
in the case of an “individual measurements” chart, the order
of the measurements.) Subgrouped measurements in a
DATA= data set must be arranged in strung-out form (one
measurement per observation).

�Although the discussion here refers to control charts for
variables (continuous measurements), similar guidelines
apply in the case of control charts for attributes.

Examples: The SAS data set CHEMICAL listed in Output 1
is read as a DATA= input data set because it contains a
process variable (XT) and a subgroup variable (T). The
following statements create a chart for individual measure-
ments and moving ranges (not shown):

proc shewhart data=chemical graphics;
irchart xt * t;

run;

Similarly, the SAS data set FILM listed in Output 6 is read
as a DATA= input data set because it contains a process
variable (TESTVAL) and a subgroup variable (SAMPLE).
Each subgroup consists of four consecutive measurements
with the same value of SAMPLE. The following statements
create an X and R chart (not shown):

proc shewhart data=film graphics;
xrchart testval * sample;

run;

HISTORY= Data Sets for Reading Summary Data

In some applications, your data may consist of subgroup
means and ranges (or standard deviations) rather than raw
measurements. You use a HISTORY= input data set to
provide the SHEWHART procedure with summary data of
this form.

The variables that you include in a HISTORY= data set
depend on the chart statement that you are using. For
instance, if you are using the XRCHART statement, your
HISTORY= data set must include four variables: a subgroup
variable, a subgroup mean variable, a subgroup range
variable, and a subgroup sample size variable. The names
of the last three variables must begin with a common prefix
and end with the suffix characters X , R, and N, respectively.

Example. Suppose that a SAS data set named MACHINE
contains four variables: a subgroup variable named TIME,
a mean variable named VALUEX, a range variable named
VALUER, and a sample size variable named VALUEN.
Then the following statements create an X and R chart:

proc shewhart history=machine graphics;
xrchart value * time;

run;

Note that the prefix VALUE is specified in the XRCHART
statement.

None of the earlier examples in this paper use HISTORY=
input data sets, but examples are provided in SAS Institute
Inc. (1989b, 1994).

You should use HISTORY= data sets in applications where
the data arrive in summarized form (for instance, from a
data collection device). You can use a variety of SAS
summarization procedures to create a HISTORY= data
set, and you can also create a HISTORY= data set as a
SHEWHART output data set by using the OUTHISTORY=
option in any chart statement.

LIMITS= Data Sets for Reading Control Limit Informa-
tion

You use a LIMITS= data set to provide the SHEWHART
procedure with either a complete set of pre-computed fixed
control limits or a set of parameters from which the displayed
control limits are to be computed. You can specify a LIMITS=
data set in conjunction with a DATA= or HISTORY= data
set. If you do not specify a LIMITS= data set, the control
limits are computed from the data.

Each observation in a LIMITS= data set provides a set of
control limits (or parameters) to be displayed with a BY
group of observations for a particular process variable. The
variables that you provide in a LIMITS= data set depend
upon the chart statement that you are using, and they have
reserved names beginning and ending with the underscore
character. You can create a LIMITS= data set with a DATA
step program, or you can create a LIMITS= data set as
a SHEWHART output data set by using the OUTLIMITS=
option in any chart statement. The latter method is partic-
ularly useful if you compute startup limits to be saved for
monitoring new data.

Examples. The data set DELAYLIM listed in Output 17
can be read as a LIMITS= data set to provide a full set of
pre-determined individual measurement and moving range
control limits for a process variable named TIME. The fol-
lowing statements create this chart (not shown):

proc shewhart data=calls
limits=delaylim graphics;

irchart time * recnum / readlimits;
run;

Note that you specify the READLIMITS option to indicate
that the control limits are to be read from the LIMITS= data
set rather than computed from the data.�

In contrast, the data set DIFFLIM listed in Output 13 does
not contain a full set of control limits. However, it can be
read as a LIMITS= data set because it provides control limit
parameters (estimates of the process mean and standard
deviation) from which individual measurement and moving
range control limits can be computed for a process variable
named DIFF. The following statements create this chart
(see Figure 13):

proc shewhart data=new
limits=difflim graphics;

irchart diff * sample / readlimits;
run;

Note that the equations used to compute the control limits
from the specified parameters depend on the chart state-
ment that you are using. For details of these equations,
refer to SAS Institute Inc. (1989a, 1994).

�In Release 6.10, the READLIMITS option is not required
for this purpose, and control limits are read from the LIMITS=
data set if one is specified. The READLIMITS option is
redundant, and you can specify NOREADLIMITS to override
the new default.

TABLE= Data Sets for Creating Customized Charts

Unlike a LIMITS= data set, which is useful in situations
where you want to display pre-determined limits for standard
control charts, a TABLE= input data set is appropriate for
non-standard applications where you want to display highly
customized, pre-computed chart statistics and control limits.

Each block of observations in a TABLE= data set provides
a set of chart statistics and control limits to be displayed for
a particular process variable. These blocks are identified
by a variable named VAR whose values are the process
variables specified in the chart statement. The particular
variables that you provide in a TABLE= data set depend
upon the chart statement that you are using, and they have
reserved names beginnning and ending with the underscore
character. A TABLE= data set must also include a subgroup
variable.

You can create a TABLE= data set with a DATA step
program, and you can also create a TABLE= data set as
a SHEWHART output data set by using the OUTTABLE=
option in any chart statement.

Example. In the section on “Autocorrelation in Process
Data”, the data set EWMATAB is read as a TABLE= in-
put data set. A partial listing of EWMATAB is shown in
Output 20.

Output 20. Partial Listing of EWMATAB

_VAR_ T _LCLX_ _SUBX_ _MEAN_ _UCLX_ _SIGMAS_ _LIMITN_ _SUBN_

xt 2 84.2620 90 96.0000 107.738 3 1 1
xt 3 79.2722 89 90.8825 102.493 3 1 1
xt 4 77.6755 89 89.2830 100.890 3 1 1
xt 5 77.4351 94 89.0426 100.650 3 1 1
xt 6 81.6469 91 93.2543 104.862 3 1 1
xt 7 79.7317 89 91.3391 102.947 3 1 1
xt 8 77.7444 93 89.3518 100.959 3 1 1
. . . . . . . . .
. . . . . . . . .

xt 98 71.0715 83 82.6790 94.286 3 1 1
xt 99 71.3443 86 82.9517 94.559 3 1 1
xt 100 73.9341 88 85.5415 97.149 3 1 1

The variables LCLX , MEAN , and UCLX provide the
control limits, and the variable SUBX provides the chart
statistic. The variables LIMITN and SUBN provide the
nominal sample size associated with the control limits and
the subgroup sample size (these are not necessarily the
same in general.) The constant value of the variable
SIGMAS indicates that the control limits are to be labeled

as 3� limits.

The following statements display the chart specified by
EWMATAB (see Figure 4):

proc shewhart table=ewmatab graphics;
xchart xt * t;

run;

Although the XCHART statement is used here, the chart is
not a Shewhart X chart because the variables in EW-
MATAB were created by the ARIMA procedure, and the
SHEWHART procedure displays their values without inter-
vention.

Printed in the U. S. A.

